

PRACTICAL SESSION FOR “ADVANCED IMAGE ANALYSIS”

D. Legland

August 11, 2023

Abstract *This document is the guideline for the practical session about “advanced image analysis”. It is composed of two parts that follow the talk outline. In the first part, it is proposed to investigate errors resulting from image analysis algorithms. The second part is devoted to image texture analysis. Both parts can be treated independently.*

Contents

1	Validation of image analysis	2
1.1	Requirements	2
1.2	Comparison of perimeter measures	2
1.3	Average error	5
1.4	Impact of orientation	6
1.5	Extension to 3D	8
2	Image texture analysis	9
2.1	Requirements	9
2.2	Basic image	9
2.3	Cellular morphology of plant tissue	10
2.4	Anisotropic image	11
2.5	Collection of images	12

1 Validation of image analysis

This practical proposes to investigate errors made by image analysis algorithm, and to compare the behaviour of several algorithms.

1.1 Requirements

The practical will use the Fiji software. The following plugins are necessary:

- MorphoLibJ (installed by default)
- Digital_Shapes¹, a library that allows generating binary images of discretized shapes in 2D and 3D.
- ij_Geometry², a geometry processing library for ImageJ, required by Digital_Shapes.

A basic knowledge of ImageJ and of macro will be necessary.

1.2 Comparison of perimeter measures

In this session, we will compare the results of several methods for measuring the perimeter of binary shapes. We will focus on basic shapes with known theoretical values of perimeter to be able to evaluate resulting errors, and compare the results obtained with native ImageJ's "Analyze Particles" and the "Analyze Regions" from the MorphoLibJ library.

We will evaluate (relative) errors based on the following equation:

$$\delta_p^{\text{method}} = \frac{P_{\text{method}} - P_{th}}{P_{th}} \quad (1)$$

1.2.1 Disk

Let us start with the most simple shape: the disk. The perimeter of a disk with radius r is $2\pi r$. When using a disk with radius 50 pixels, the expected value of perimeter is around $P_{th} = 314.15$.

- Create a new image, with size 200x200
- Run the Plugins ▸ Digital Shapes ▸ Fill Disk
- Choose a center around (100, 100), a radius equal to 50, and click "OK"
- Note that you can slightly shift the center to make the disk less "symetric". Use for example (100.23, 100.34).

You should obtain a disk in the middle of your image (Fig. 1).

¹<https://github.com/ijtools/ijDigiShapes>

²<https://github.com/ijtools/ijGeometry>




Figure 1: Examples of discretized disks with same radius ($R=20$). The left-most image corresponds to a disk centered with pixel grid.

Result with ImageJ

You can measure perimeter with standard functions of ImageJ via the “Analyze Particles” plugin. You may need to set the threshold before (Image ▸ Adjust ▸ Threshold..., click “yes”), and to make sure the perimeter is selected in the list of measures (Analyze ▸ Set Measurements, select “Perimeter”). Check the value of perimeter. Evaluate error as

$$\delta_P^{IJ} = \frac{P_{IJ} - P_{th}}{P_{th}}$$

Result with MorphoLibJ

Compare the error with that obtained with MorphoLibJ. Use the Plugin “Plugins ▸ MorphoLibJ ▸ Analyze ▸ Analyze Regions...”, and make sure the option for perimeter is checked. Evaluate error as

$$\delta_P^{MLJ} = \frac{P_{MLJ} - P_{th}}{P_{th}}$$

One should obtain a smaller relative error with MorphoLibJ. Note that relative error depends on shape and on relative position and orientation of shape with respect to discretization grid.

Shape	P_{th}	ImageJ		MorphoLibJ	
		P	δ_P	P	δ_P
disk $R=20$	125.66	130.1	3.55%	125.0	-0.5%
disk $R=50$	314.15	329.7	4.95%	314.2	0.03%

Table 1: Sample results that can be obtained for perimeter of a disk.

Comparisons between both methods on a sample image are reported in Table 1. Note that the results depends on the position of the center of the shape.

1.2.2 Other shapes

You can repeat the workflow with other shapes: square, ellipses, rotated rectangles... It can be interesting to let vary the different shape parameters, as well as the orientation.





(a) Ellipse, $a = 20, b = 10$ (b) square, $s = 30$ (c) box, $s_1 = 40, s_2 = 20$ (d) capsule, $l = 30, t = 20$

Figure 2: Examples of discrete shapes.

Note that the computation of ellipse perimeter requires numerical integration. Expected perimeter values for disks and ellipses with various sizes are given in table 2.

Radius	Perimeter	Radius1	Radius 2	Perimeter
10	62.83	20	10	96.88
20	125.66	30	10	133.65
30	188.49	40	20	193.77
40	251.33	50	10	210.10
50	314.15	50	20	230.13
		50	30	283.62

(a) Disks

(b) Ellipses

Table 2: Expected perimeter values for disks and ellipses with various radiusses.

In order to facilitate repetitive runs, a macro can be used. Run “Plugins New Macro” to open the macro editor, and choose the language of your choice. The following script³ provides a macro that can be used to run the analysis on an ellipse by randomizing position and orientation.

```

1 // shape parameters
2 r1 = 40;
3 r2 = 20;
4 perimTh = 193.77;
5
6 // check with fixed orientation
7 imgEllipse = "ellipse40x20";
8 newImage(imgEllipse, "8-bit_black", 200, 200, 1);
9 centerX = 100 + random;
10 centerY = 100 + random;
```

³https://github.com/dlegland/image-analysis-practical-2023/blob/main/practical/validation/macros/ellipse_perimeter.ijm

```

11  theta = random * 180;
12  run("Fill_Ellipse", "center_x=&centerX_center_y=&centerY_major=&r1_minor=&r2_orientation=&theta_fill
    =255");
13
14  // Run analysis with ImageJ
15  setAutoThreshold("Default_dark");
16  run("Analyze_Particles...", "display_clear");
17
18  // retrieve perimeter and compute error
19  perimIJ = Table.get("Perim.", 0);
20  errorIJ = 100 * (perimIJ - perimTh) / perimTh;
21  print("Relative_Error_using_ImageJ:_" + errorIJ + "%");
22
23  // Run analysis with MorphoLibJ
24  run("Analyze_Regions", "area_perimeter_equivalent_ellipse");
25  selectWindow(imgEllipse + "Morphometry");
26
27  // retrieve perimeter and compute error
28  perimMLJ = Table.get("Perimeter", 0);
29  errorMLJ = 100 * (perimMLJ - perimTh) / perimTh;
30  print("Relative_Error_using_MorphoLibJ:_" + errorMLJ + "%");

```

1.3 Average error

It may be interesting to check the variability of the measures around the theoretical values or/and the average values provided by the plugins. For this, we can generate a collection of images representing shapes with same size parameters, but different positions (and orientations if relevant).

The best is to use macro programming. An example is given below⁴, for rotated ellipses with semi-axis lengths equal to 40 and 20.

```

1  // shape parameters
2  r1 = 40;
3  r2 = 20;
4  perimTh = 193.77;
5
6  // the image name
7  imgName = "ellipse40x20";
8
9  // number of repetitions
10 nRepets = 10;
11
12 // initialize array
13 perimIJArray = newArray(nRepets);
14 perimMLJArray = newArray(nRepets);
15
16 run("Clear_Results");
17
18 for (i=0; i < nRepets; i++) {
19     // check with fixed orientation
20     newImage(imgName, "8-bit_black", 200, 200, 1);
21     centerX = 100 + random;
22     centerY = 100 + random;
23     ori = random * 180;

```

⁴https://github.com/dlegland/image-analysis-practical-2023/blob/main/practical/validation/macros/ellipse_perimeter_variability.ijm

1 Validation of image analysis

```
24     run("Fill_Ellipse", "center_x=&centerX_center_y=&centerY_major=&r1_minor=&r2_orientation=&ori_fill  
      =255");  
25  
26     // Run analysis with ImageJ  
27     setAutoThreshold("Default_dark");  
28     run("Analyze_Particles...", "display");  
29  
30     // retrieve perimeter and compute error  
31     selectWindow("Results");  
32     perimIJ = Table.get("Perim.", i);  
33     print(perimIJ);  
34     perimIJArray[i] = perimIJ;  
35     errorIJ = 100 * (perimIJ - perimTh) / perimTh;  
36     print("Relative_Error_using_ImageJ:_" + errorIJ + "%");  
37  
38     // Run analysis with MorphoLibJ  
39     run("Analyze_Regions", "area_perimeter");  
40     selectWindow(imgName + "-Morphometry");  
41  
42     // retrieve perimeter and compute error  
43     perimMLJ = Table.get("Perimeter", 0);  
44     perimMLJArray[i] = perimMLJ;  
45     errorMLJ = 100 * (perimMLJ - perimTh) / perimTh;  
46     print("Relative_Error_using_MorphoLibJ:_" + errorMLJ + "%");  
47     // clean up  
48     close(imgName);  
49 }  
50  
51 Array.getStatistics(perimIJArray, min, max, mean, std);  
52 print("ImageJ_stats");  
53 print("___min:_" + min);  
54 print("___max:_" + max);  
55 print("___mean:_" + mean);  
56 print("___std_dev:_" + std);  
57  
58 Array.getStatistics(perimMLJArray, min, max, mean, std);  
59 print("MorphoLibJ_stats");  
60 print("___min:_" + min);  
61 print("___max:_" + max);  
62 print("___mean:_" + mean);  
63 print("___std_dev:_" + std);
```

1.4 Impact of orientation

Choose an oriented box, with fixed side lengths, and generate discrete versions of the shape with various orientations.

The best is to create a macro. An example of expected result (for a rotated rectangle / oriented box with side length 150 and 50) is given in Figure 3). The corresponding macro can be found online⁵.

```
1 // shape parameters  
2 sidel = 150;  
3 side2 = 50;  
4 perimTh = 400;  
5 centerX = 100 + random;
```

⁵https://github.com/dlegland/image-analysis-practical-2023/blob/main/practical/validation/macros/rectangle_perimeter_byTheta.ijm




Figure 3: Impact of orientation for computing perimeter of a rotated rectangle

```

6  centerY = 100 + random;
7  // the image name
8  imgName = "ellipse40x20";
9  // number of repetitions
10 nTheta = 180;
11 // initialize arrays
12 thetaArray = newArray(nTheta);
13 perimIJArray = newArray(nTheta);
14 errorIJArray = newArray(nTheta);
15 perimMLJArray = newArray(nTheta);
16 errorMLJArray = newArray(nTheta);
17 run("Clear_Results");
18
19 for (i=0; i < nTheta; i++) {
20   // check with fixed orientation
21   newImage(imgName, "8-bit_black", 200, 200, 1);
22   theta = i;
23   run("Fill_Oriented_Box", "center_x=&centerX_center_y=&centerY_box_length=&side1_box_width=&side2_
      orientation=&theta_fill=255");
24   thetaArray[i] = theta;
25
26   // Run analysis with ImageJ
27   setAutoThreshold("Default_dark");
28   run("Analyze_Particles...", "display");
29
30   // retrieve perimeter and compute error
31   selectWindow("Results");
32   perimIJ = Table.get("Perim.", i);
33   print(perimIJ);
34   perimIJArray[i] = perimIJ;
35   errorIJ = 100 * (perimIJ - perimTh) / perimTh;
36   errorIJArray[i] = errorIJ;
37
38   // Run analysis with MorphoLibJ
39   run("Analyze_Regions", "area_perimeter");
40   selectWindow(imgName + "-Morphometry");
41

```

1 Validation of image analysis

```
42 // retrieve perimeter and compute error
43 perimMLJ = Table.get("Perimeter", 0);
44 perimMLJArray[i] = perimMLJ;
45 errorMLJ = 100 * (perimMLJ - perimTh) / perimTh;
46 errorMLJArray[i] = errorMLJ;
47 close(imgName);
48 }
49
50 Plot.create("Variations_of_Perimeter", "Rectangle_Orientation", "Perimeter");
51 Plot.setLimits(0, 180, 370, 435);
52 Plot.setLineWidth(2);
53 Plot.setColor("red");
54 Plot.add("line", thetaArray, perimIJArray);
55 Plot.setColor("blue");
56 Plot.setLineWidth(2);
57 Plot.add("line", thetaArray, perimMLJArray);
58 Plot.setLegend("ImageJ\\tMorphoLibJ", "top-right");
```

1.5 Extension to 3D

The same workflow can be applied to 3D images as well. The “Digital Shapes” library provides discretization of various 3D shape models: ellipsoids, cuboids, cylinders, capsules... The computation may be somewhat slower however.

2 Image texture analysis

This practical session proposes to investigate image texture analysis using gray level granulometry with mathematical morphology filters.

2.1 Requirements

The practical will use the Fiji software. The following plugins are necessary:

- MorphoLibJ (installed by default)
- ijGranulometry⁶ a grayscale granulometry plugin for ImageJ based on MorphoLibJ

2.2 Basic image

This sections presents the basic of granulometry.

We will start with the image “coins.png” (Fig. 4). This image presents coins with various diameters.




Figure 4: Test image for granulometry

Run the plugin “Granulometry > Granulometry by Radius...”, fill in the dialog with appropriate settings:

- as the coins are bright over a dark background, choose “opening” operation
- use disk structuring element to match the size of the coins
- choose a max radius larger than that of the coins, say 40 pixels
- for large images, using a step larger than 1 reduces computation time.
- for this image, leave calibration as is.

⁶<https://github.com/ijtools/ijGranulometry>

2 Image texture analysis

The resulting granulometry curve is presented on Fig. 5. The abscissa presents the diameter of the structuring element (computer as $2 \cdot \text{radius} + 1$). Two peaks can be noticed around 50 and 60 pixels. You may check that this corresponds to the diameter of the coins by using the “Plot profile” tool.




Figure 5: Result of granulometry Plugin on coins image.

If the “Display Volume Curve” option was selected, the sum of gray levels for each opening result is also displayed. Two “shoulders” may be noticed, that correspond to the diameters when large regions disappear with opening.

2.3 Cellular morphology of plant tissue

We will now investigate cellular morphology of plant tissue based on cross section of maize stems observed with macroscopy imaging. Open one of the images in the “maize” directory (Fig. 6). We can observe many polygonal cells, as well as bright structures (vascular bundles).




Figure 6: Application of gray-scale granulometry on an image of plant tissues.

We run the “Granulometry by Radius...” plugin as follow:

- as cells are dark and separated by bright cell walls, we will use the “closing” operation
- use disk structuring element
- choose a max radius larger than that of the cells, say 80 pixels
- use a step equal to 1
- leave calibration as is

The result of granulometry presents a large peak around 30 pixels, corresponding to the typical diameter of cells.

2.4 Anisotropic image

Granulometry may also be used to identify preferred orientations within images. Let us open the image “f7a1.tif” image in the tomato directory. The image represents a slice of tomato pericarp, obtained from macroscopy imaging.




Figure 7: Sample image of tomato slice.

Let us apply granulometry, and compare the results obtained when changing the shape of the structuring elements:

- use the “Closing” operation
- repeat the process with each one of the following shapes: “square”, “horizontal line”, and “vertical line”.

2 Image texture analysis

- choose a max radius “larger enough”, say 80 or 100 pixels
- use a step equal to 1 or larger to reduce computation time.
- leave calibration as is

The three resulting curves have different shapes. The curve obtained with a square or an horizontal line presents a thinner distribution around 40 pixels. This corresponds to variation around the typical thickness of cells. Using a vertical linear structuring elements results in a thicker bulb, and an shift to the right. This may be interpreted as a larger variability in cell sizes in the vertical direction, and as a larger typical size in the vertical direction.




Figure 8: Granulometry curves on image of tomato tissue using different shapes of structuring elements.

2.5 Collection of images

When working with a large number of images, it may be tedious to repeat the process of running the granulometry analysis. One possibility is to run a macro. Another possibility is to run the “batch Granulometry” plugin. The plugins allows to select a directory, the granulometry analysis options, and to perform the same analysis on the whole set of images within the directory.